Parsing Idris

A part of the Tools and Tactics for Idris project

Ahmad Salim Al-Sibahi (asal@itu.dk)

IT University of Copenhagen
Supervisors:
David Raymond Christiansen (drc@itu.dk)
Dr. Peter Sestoft (sestoft@itu.dk)

December 11, 2013

1 Introduction

Idris [1] is a strict pure dependently-typed programming language inspired by Haskell [2] and ML-style languages [3]. As a dependently-typed language Idris allows types to be indexed by values e.g. the natural numbers, in addition to being parametrised over other types like other functional languages. This makes Idris suitable for reasoning formally about programs, and makes it suitable for a variety of applications such as writing embedded domain specific languages (EDSLs) because many errors can be reported at compile time.

Yet all this flexibility comes at a price. Since there is not distinction between types and terms, the grammar is vastly complicated. Moreover, the focus on EDSLs has brought in more features such as syntactic extensions, which makes reasoning about the complete grammar impossible statically. Partly because of this and partly because of the indentation sensitive Haskell-like syntax; Idris is not a particularly trivial language to parse.

One must therefore carefully evaluate the way the grammar is written, and write the parser in such way that there is little room for ambiguity. Furthermore, the source code must be tracked carefully in order to provide sensible errors in all parts of the compiler pipeline, and that external tools can use this to provide additional features such as clickable source code.

A parser written in the Parsec [4] library already exist for Idris. Yet because of all the advanced features of Idris, and the lack of time spent on the parser to accommodate these features, error locations and messages are highly inaccurate.

In this paper I will outline my work regarding restructuring of the Idris parser such that the grammar is more sensible, the parser provides better errors and that the source code is tracked more accurately. In Section 2 I will discuss the Idris grammar and what particular difficulties there can be when trying to parse it. Section 3 discusses the particular parser technology used, and the advantages of using it. Section 4 describes the specific effort put into improving the grammar and parser of Idris. I will evaluate the solution in Section 5 and discuss future work in Section 6. Lastly I conclude in Section 7.

2 Idris Syntax

The syntax of Idris is heavily inspired by the syntax of Haskell; with some modifications to better accommodate the dependent types-based programming paradigm and EDSLs.

In this section I will use examples to highlight some of the features that makes creating a parser non-trivial.

Figure 1 showcases a simple inductive data declaration for natural numbers, and a pattern-matching definition for a function 'add' which computes the sum of two naturals. The only noticeable difference in syntax from an ordinary Haskell program is that ':' is used for type declarations instead of '::'.

```
data N =
    zer
  | suc N

add : N -> N -> N
add zer     m = m
add (suc n) m = suc (add n m)
```

Figure 1: Simple Idris program

Similarly to Haskell, Idris allows one to define custom operators with user-defined fixity and precedence. This can be seen in Figure 2 where the data type for existential types is using the operator '`***`' as a name.

Another thing the reader might notice in Figure 2 is that both the '`***`' operator and a lambda declaration were allowed in the type signature for '`take_while`'.

Lastly, the reader should notice that much of the syntax presented is indentation-sensitive; e.g. the data declarations and the '`where`'-block.

```
infix 8 ***

data (***) : (a : Type) -> (b : a -> Type) -> Type where
  exists : {a : Type} -> {b : a -> Type} -> (w : a) -> (p : b w) -> a *** b

infixr 5 :::

data Vector : (n : N) -> (a : Type) -> Type where
  nil   : {a : Type}                                -> Vector zer a
  (:::) : {n : N} -> {a : Type} -> a -> Vector n a -> Vector (suc n) a

take_while : (p : a ->  Bool) -> Vector n a -> N *** (\ m => Vector m a)
take_while p nil       = exists zer nil
take_while p (x ::: xs) with (take_while p xs)
  | exists p ys = if p x then exists (suc m) (x ::: ys) else eps
  where eps = exists zer nil
```

Figure 2: Operators and lambdas

For improved readability of EDSLs, Idris allows further expansion of the syntax. In Figure 3 there has been defined two syntactic extensions, one for the '`take_while`' function and one for the existential data type. The syntax definitions (line 1-2) can define new keywords or syntactic markers (specified as either simple identifiers or quoted in a string), capture other expressions which are delimited in square brackets (e.g. '`[vect]`' in the example), or capture names which are delimited in curly braces (e.g. '`{x}`') that can be used for binding. This makes syntactic extensions quite powerful, and the only thing one cannot extend the language with is recursive grammar definitions.

At lines 10-11 in Figure 3 there is a clause which uses these syntactic extensions as any other built-in syntax. Rather than being baked into Idris, many things like '`if`'-expressions are in practice defined using syntax extensions in the library.

```
syntax take' [vect] as {x} while [body] = take_while (\x => body) vect
syntax "∃" {x} "→" [ty] = _ *** (\x => ty)

isEven : N → Bool
isEven zer     = True
isEven (suc x) with (isEven x)
  | True  = False
  | False = True

testSyntax : ∃ m → Vector m N
testSyntax = take' suc (suc zer) ::: zer ::: suc zer ::: nil as i while isEven i
```

Figure 3: Syntax extensions

3 Trifecta

Trifecta [5] is a monadic parser combinator library created by Edward Kmett, with focus on providing good tools for error reporting and incremental parsing.

In this section I will explain what parser combinators are, and how Trifecta as a library is designed such that the combinators provide the aforementioned features.

3.1 Parser Combinators

Traditionally there are two ways of creating a parser: either by hand-writing one, or by giving the grammar as input to a parser generator such as yacc [6] and generating one.

By now a common way of parsing in the language community is using parser combinator libraries. Parser combinator libraries [7, Chapter 16] [8] assist in creating hand-written parsers by providing, often in the form of an EDSL, a set of generic higher-order functions called parser combinators. Parser combinators produce new parsers by taking other parsers as input, and provide features such as alternation, sequencing, choice, repetition and cursor movement.

3.1.1 Combinators by example

```
landCode = string "+45"

phoneNumber = option "" landCode *> count 8 digit

username = many alphaNum

login = do id <- try phoneNumber <|> username
           string "--"
           password <- some alphaNum
           return (id, password)
```

Figure 4: Parsing a login

Figure 4 shows a simple example of a parser for parsing login information given in the following format "id–password". In this format id is either a phone number or a self-chosen user-name and password is a string of alphanumerical characters.

The choice for id is created by using the '`<|>`' combinator, which tries parsing the first alternative and if it fails, without consuming any input, it will try

parsing the second alternative. Since there is overlap between the definition of phone number and user-name parser, the phone number parser must be wrapped in a '`try`'-combinator which ensures that input can be restored by backtracking such that no input is consumed and other alternatives can be tried. As we will see in Section 4 the '`try`'-combinator is double-edged sword, permitting easier parsing of the ambiguous elements in the grammar but at the price of bad performance and error messages. with the price

Because the parser is monadic, sequencing multiple parsers is conveniently available using do-notation.

In the example there are three repetition parser combinators used: 'count' which specifies an exact number of repetitions, 'many' which specifies zero or more repetitions and 'some' which specifies one or more repetitions; although many more are usually available in such libraries.

Finally, one can observe that the phone number parser uses an combinator 'option' which allows specification of an optional part of the input and the combinator '`*>`' which ignores the result of the left operand and returns the result of the right operand.

3.2 Features of Trifecta

While there are many parser combinator libraries for Haskell, Trifecta provides the following desirable features:

Incremental parsing When working with external tools the input might change quite often. It is therefore desirable if only the relevant part of the input is parsed again, and Trifecta achieves this by using monoidal parsing [9]. This feature provide baseline support for semantic highlighting, code completion and other types of parser-dependent tooling.

Error reporting Quality diagnostics and pretty printing are a key design goal for Trifecta. If a parser fails, it doesn't just show what it expected but also nicely formats and points to the place of error. Moreover, Trifecta provides utilities to highlight semantic errors in the code using the same type of pretty printing it uses for syntactic errors.

4 Solution

In many ways rewriting the parser from Parsec to Trifecta is non-trivial. Nonetheless, I will outline in this section my effort on improving the parser by highlighting the interesting parts of my work, namely formalization of the grammar and improvement on the error reporting and locating.

4.1 Formalising the Grammar

Part of the challenge of writing a new parser for Idris is that, at the time, there was no formalization of the grammar of Idris. This caused two issues:

1. There was no official reference regarding the Idris syntax. This meant that it was hard to find the syntactically correct way to write a program and in case of error one could either:

- try to isolate the bug in a structured fashion and try to guess how it should be corrected, or
- laboriously time analyse the parser code and see what the legal syntax should have been

2. There were some grammatical inconsistencies, such that some programs were improperly accepted (e.g. programs that were missing a '}') and some didn't behave as expected (e.g. 'proof'-blocks worked differently from 'do'-blocks).

The formalization to an official BNF grammar was further complicated by the fact that Idris supports an indentation-based syntax and user defined extensions. As such the resulting grammar is incomplete since it is impossible to formalize all possible syntax extensions and it does not reflect all of the lexical properties of indentation. The latter could be improved by specifying the grammar in an extended BNF syntax like the one used by Adams [10], but at the time of analyses there were too many inconsistencies regarding indentation in order to completely reflect the actual syntax and thus this was omitted. In any case the grammar should provide a useful, covering and understandable overview of the syntax.

The actual translation process was done by carefully analysing the existing parser structure, noting any seeming inconsistencies and correcting them in the grammar. The final result is available in Appendix A.

4.2 More Informative Failures

Another issue there was with the existing parser was that grammar rules were improperly documented. Instead of error messages showing what legal types of grammar it expected, it showed a list of tokens that could appear somewhat random to the intended user. For example, if a user had mistyped something in the start of an expression, it would show all operators, keywords and marker symbols that were possible at that point instead of just showing that an expression was expected.

Therefore some of the work done on the parser was to provide a human-readable string explanation for each grammar rule using the '<?>' combinator. Figure 5 shows an example of such annotation.

```
   {– | Parses an accessibilty modifier (e.g. public, private) –}
   accessibility :: IdrisParser Accessibility
3  accessibility = do reserved "public";   return Public
                  <|> do reserved "abstract"; return Frozen
                  <|> do reserved "private";  return Hidden
6                 <?> "accessibility␣modifier"
```

Figure 5: Documenting Parser Rules

4.3 The Parser with Fear of Commitment

One of the more challenging parts of the grammar re-factoring process was that many of the rules were written in such way that they didn't commit to any of the branches in a list of alternatives. This meant that each time there was an

error in one of the alternatives, the parser had to backtrack all the way back to the starting point.

A disadvantage of the try-combinator is that it requires exponential time and space if used improperly. In addition, when all of the branches fail, it will show where it was before parsing the specific grammar rule instead of where the possible error happened. In the case of top-level declarations, this can be multiple lines and columns away from the actual location, and finding the correct location could be extremely hard for the programmer.

Figure 6 shows one of the rules, were the reader may observe that all alternatives are wrapped in the try-combinator meaning it will never commit to any alternative in case of an error.

```
decl' :: SyntaxInfo -> IdrisParser PDecl
decl' syn =    try fixity
          <|> try (fnDecl' syn)
          <|> try (data_ syn)
          <|> try (record syn)
          <|> try (syntaxDecl syn)
          <?> "declaration"
```

Figure 6: Non-committing Parser Rule

To fix this two things are usually required:

1. Reordering of rules such that less ambiguous rules comes before more ambiguous rules
2. Minimizing the span of the try-combinator such that only the ambiguous part is covered, i.e. usually until a keyword or distinct character sequence.

Figure 7 shows the committing version of the rule shown in Figure 6, where some of the alternatives has been reordered to avoid backtracking. It should be noted that this was a simple example of such re-factoring, and more complex rules such as the one for data declaration required significantly more effort. While not all parsers can be healed from the fear of commitment, these few rules come a long way!

```
decl' :: SyntaxInfo -> IdrisParser PDecl
decl' syn =    fixity
          <|> syntaxDecl syn
          <|> fnDecl' syn
          <|> data_ syn
          <|> record syn
          <?> "declaration"
```

Figure 7: Committing Parser Rule

4.4 Improving Source Code Location Tracking

The old parser annotated abstract syntax with source information, but this information was highly inaccurate and was missing column information.

In some elements like atomic identifiers or references, the location was first retrieved after the token was parsed; This meant that if there was a lot of whitespace after the reference, the location would have been at the end of the whitespace, instead of near the token. For these tokens, a correction was made

such that the location was retrieved *before* a token was parsed. This additionally had the benefit of allowing easier access to highlighting relevant identifiers by simply selecting all text until the next whitespace character. Furthermore, all the abstract syntax trees at various stages in the pipeline were altered to provide column information such that multiple items on the same line were distinguishable.

5 Evaluation

In this section I will highlight some of the improvements that were discovered and fixed in the new parser, and what challenges there still are because of the way the grammar is constructed. Additionally I will highlight the improvements in error handling and the reception by the community both qualitatively and quantitatively in terms of reported bugs. Finally I will show that the source code locating tracking is working, by creating a tool for generating a clickable version of a source code file in HTML.

5.1 Syntactic Corrections and Challenges

The following bugs in the syntax were corrected:

1. '`proof`' and '`tactics`'-blocks allow the use of the off-side rule, similarly to '`do`'-blocks
2. Documentation comments can be nested inside existing comments
3. The '`where`'-keyword is optional on instances with no methods declared
4. Tactic sequences can hold more than two tactics
5. '`using` and '`parameter`' blocks can be empty
6. Checks that parentheses and braces are balanced

The only major limitation in the current implementation of the parser is the parsing of nested parenthesized expressions, which currently takes exponential time. This is due to ambiguity in the grammar of Idris where operator slices e.g. '(\\ x)', are hard to distinguish from lambdas '(\ y => y * x)' and thus all parenthesized expression require full lookahead. To solve this issue, the grammar needs to be rewritten such that there is less ambiguity between the various kinds of parenthesised expressions.

5.2 Improvements in Error Reporting

As a showcase for error reporting improvements, I will highlight a couple of examples where a simple mistake resulted in some uninformative error report by the old parser but where the new parser shows a clear placement and presentation of the error.

```
    module ErrorReport

3   data List : Type -> Type where
      Nil :: List a
      Cons :: a -> List a -> List a
```

(a) File with error

```
"./error-report.idr" (line 3, column 11):
unexpected ':'
3  expecting "infixl", "infixr", "infix", ...
```

(b) Old parser report

```
./error-report.idr:4:8: error: expected: type signature
  Nil :: List a
3       ^
```

(c) New parser report

Figure 8: Improvements in Error Reporting

Figure 8 shows an Idris program where the user had accidentally used a double colon for data constructor type signatures similarly to Haskell type signatures, instead of the single colon used in Idris type signature. As can be observed in the figure, the old parser reports that the error is at line 3 (where type is declared, not the constructor) and suggests that it expected one token from a long list of unrelated ones.

The new parser substantially improves the error location by pointing directly at the place of error and showing a part of the code that was affected. Furthermore it only suggests that the type signature should be fixed, which is the correct error in this case.

```
    module ErrorReport

3   test : Int -> Int -> Int
    test x y
    = let z = x + y in
6   do x +
```

(a) File with error

```
"./error-report2.idr" (line 4, column 6):
unexpected "x"
3  expecting ":" or operator
```

(b) Old parser report

```
./error-report2.idr:7:1: error: unexpected
    EOF, expected: expression
3  <EOF>
   ^
```

(c) New parser report

Figure 9: Improvements in Error Reporting, cont.

Figure 9 demonstrates another improved category of error messages. The user had accidentally forgot to finish an expression, stopping at an infix operator. Instead of complaining that it was missing the second operand; it instead

complains at the start of the function clause saying that it didn't expect the argument given.

Contrarily, the new parser correctly highlights that the parser didn't expect end-of-file but an expression and the user can immediately correct this bug instead of hunting down the error at the completely wrong place.

5.3 Community Reception

The parser described in this paper has officially been included in the Idris language, and is now the solely used parser. The reception by the community has been generally positive and in the word of the language designer, it has been described as following in the changelog: "New parser implementation with more precise errors."

Initially there were less than 10 bugs reported, many of which already existed in the previous parser, but were undiscovered due to bad error messages. However, the number of bugs reported regarded parsing has decreased substantially and the last reported bug was reported more than a month ago at the time of writing this report.

5.4 Clickable Source Code

To test the accuracy of the source code location tracking, a simple tool for creating clickable HTML from a single Idris module was made [11]. The tool works by parsing the source code of a module using the ordinary Idris parser, and then saving the location of all declarations in a map. Afterwards, it finds all references in the source code and inserts a link that refers back to the relevant declaration in the map.

Due to time constraints, the tool was not made to work across modules and some terms were not thoroughly analysed. However, the important part which was showing that the source code referred to the right locations did as expected and such I evaluate that the source code tracking is sufficiently capable for implementation of future tools.

6 Future Work

While there has been a lot of effort in improving the parser, not all of the advantages of the parser technology was utilized. In future work, I would suggest that the following improvements can be made:

Spanning the source code Currently the source code is only tracked by a cursor in the abstract syntax tree, a better solution would entail tracking the full span of a structure in the abstract syntax tree. This would allow easier access to pretty printing and error highlighting the affected code.

Incremental parsing Trifecta supports incremental parsing, and this could be very useful for development of tools. As part of future work it could be of great interest to utilize this feature to allow for things like semantic highlighting and code completion.

Refactoring semantics out of the parser Currently the parser does a lot of semantic checking inside the parser structure.

For example, implicit parameters are not allowed in function arguments and the parser currently disallows writing implicit argument syntax in the type signatures where function arguments are expected.

This results in a parser error saying that it didn't expect the implicit syntax, and while this is correct it might be confusing for a new user. Many of these semantic checks should be moved out of the parser and refactored in a separate syntax checking step which provide better error report for such situations.

7 Conclusion

In this project I have successfully formalised the grammar and rewritten the parser for the Idris language. I did this by analysing the existing parser, fixing inconsistencies, and rewriting the new parser in such way that the error reporting was improved and locations were tracked better.

I also conclude that the resulting work has been received well by the community, and that it can be seen as a great success that the resulting work now forms the basis for the Idris language syntax and parsing.

Finally, while there are still improvements to be made, changing the parser technology to Trifecta has provided us with a great start for improving tooling and semantic error reporting.

References

[1] Brady, E.: Idris, a general-purpose dependently typed programming language: Design and implementation. Journal of Functional Programming **23** (9 2013) 552–593

[2] Marlow, S., et al.: Haskell 2010 language report. `http://www.haskell.org/onlinereport/haskell2010` (2010)

[3] Milner, R.: The definition of standard ML: revised. The MIT press (1997)

[4] Leijen, D., Meijer, E.: Parsec: Direct style monadic parser combinators for the real world. (2001)

[5] Kmett, E.: Trifecta official page. `http://ekmett.github.io/trifecta/`

[6] Johnson, S.C.: Yacc: Yet another compiler-compiler. Volume 32. Bell Laboratories Murray Hill, NJ (1975)

[7] O'Sullivan, B., Goerzen, J., Stewart, D.: Real World Haskell. 1st edn. O'Reilly Media, Inc. (2008)

[8] Hutton, G., Meijer, E.: Monadic parser combinators. School of Computer Science and IT, University of Nottingham (1996)

[9] Kmett, E.: Iterators, Parsec and Monoids – A Parsing Trifecta. http://comonad.com/reader/wp-content/uploads/2009/08/A-Parsing-Trifecta.pdf Accessed 2013-12-06.

[10] Adams, M.D.: Principled parsing for indentation-sensitive languages: revisiting landin's offside rule. In: Proceedings of the 40th annual ACM SIGPLAN-SIGACT symposium on Principles of programming languages, ACM (2013) 511–522

[11] Al-Sibahi, A.S.: Tool for generating clickable source code of idris files. https://github.com/ahmadsalim/Idris-dev/blob/feature/clickablesource/src/Util/Clickable.hs Accessed 2013-12-06.

A Idris Formalized Grammar

```
{- Header and Imports are currently parsed separatly -}
Main ::= ModuleHeader Import* Prog;

{-
  shortcut notation:
    ~CHARSEQ = complement of char sequence (i.e. any character except CHARSEQ)
    RULE? = optional rule (i.e. RULE or nothing)
    RULE* = repeated rule (i.e. RULE zero or more times)
    RULE+ = repeated rule with at least one match (i.e. RULE one or more times)
    RULE! = invalid rule (i.e. rule that is not valid in context, report meaningful error in case)
    RULE{n} = rule repeated n times
-}

EOL_t ::= '\n'  |EOF_t;

StringChar_t* = {- Any valid Haskell string character or escape code -};
StringLiteral_t ::= '"' StringChar_t* '"';

DocCommentMarker_t ::= '|' | '^';

DocComment_t ::=   '--' DocCommentMarker_t ~EOL_t* EOL_t
                 | '{-' DocCommentMarket_t ~'-}'* '-}'
                 ;

SingleLineComment_t ::= '--' EOL_t
                  |     '--' ~DocCommentMarker_t ~EOL_t* EOL_t
                  ;

MultiLineComment_t ::=
  '{--}'
  | '{-' ~DocCommentMarker_t InCommentChars_t
  ;

InCommentChars_t ::=
  '-}'
  | MultiLineComment_t InCommentChars_t
  | ~'-}'+ InCommentChars_t
  ;


Whitespace_t ::=
  SimpleWhitespace_t
  | SingleLineComment_t
  | MultiLineComment_t
  ;

Identifier_t ::= ['a'-'z''A'-'Z']['a'-'z''A'-'Z''0'-'9''_''.']+;
Operator_t ::= [':''!''#''$''%''&''*''+''.''/''<''=''>''?''@''\\''^''|''-''~']+

ModuleHeader ::= 'module' Identifier_t ';'?;

Import ::= 'import' Identifier_t ';'?;

Prog ::= Decl* EOF;

Decl ::=
    Decl'
  | Using
  | Params
  | Mutual
  | Namespace
  | Class
  | Instance
  | DSL
  | Directive
  | Provider
  | Transform
  | Import!
  ;

Decl' ::=
    Fixity
  | FnDecl'
  | Data
  | Record
  | SyntaxDecl
  ;

SyntaxDecl ::= SyntaxRule;

SyntaxRuleOpts ::= 'term'  |'pattern';

SyntaxRule ::=
  SyntaxRuleOpts? 'syntax' SyntaxSym+ '=' TypeExpr Terminator;

SyntaxSym ::=   '[' Name_t ']'
              |  '{' Name_t '}'
              |  Name_t
              |  StringLiteral_t
              ;

FnDecl ::= FnDecl';

{- NOTE: Check compatible options -}
FnDecl' ::=
```

```
    DocComment_t? FnOpts* Accessibility? FnOpts* FnName TypeSig Terminator
  | Postulate
  | Pattern
  | CAF
  ;

Postulate ::=
  DocComment_t? 'postulate' FnOpts* Accesibility? FnOpts* FnName TypeSig Terminator
  ;

Using ::=
  'using' '(' UsingDeclList ')' OpenBlock Decl+ CloseBlock
  ;

Params ::=
  'parameters' '(' TypeDeclList ')' OpenBlock Decl+ CloseBlock
  ;

Mutual ::=
  'mutual' OpenBlock Decl+ CloseBlock
  ;

Namespace ::=
  'namespace' identifier OpenBlock Decl+ CloseBlock
  ;

Fixity ::=
  FixityType Natural_t OperatorList Terminator
  ;

FixityType ::=
  'infixl'
  | 'infixr'
  | 'infix'
  | 'prefix'
  ;

OperatorList ::=
  Operator_t
  | Operator_t ',' OperatorList
  ;

MethodsBlock ::= 'where' OpenBlock FnDecl* CloseBlock
           ;

Class ::=
  DocComment_t?  Accessibility? 'class' ConstraintList? Name ClassArgument* MethodsBlock?
  ;

ClassArgument ::=
   Name
   | '(' Name ':' Expr ')'
   ;

Instance ::=
  'instance' InstanceName? ConstraintList? Name SimpleExpr* MethodsBlock?
  ;

InstanceName ::= '[' Name ']';

FullExpr ::= Expr EOF_t;

Expr ::= Expr';

Expr' ::= {- External (User-defined) Syntax -}
      |   InternalExpr;

InternalExpr ::=
  App
  | MatchApp
  | UnifyLog
  | RecordType
  | SimpleExpr
  | Lambda
  | QuoteGoal
  | Let
  | RewriteTerm
  | Pi
  |DoBlock
  ;

Name ::= IName_t;

OperatorFront ::= (Identifier_t '.')? '(' Operator_t ')';

FnName ::= Name  |OperatorFront;

Accessibility ::= 'public' | 'abstract' | 'private';

FnOpts ::= 'total'
  | 'partial'
  | 'implicit'
  | '%' 'assert_total'
  |'%' 'reflection'
  | '%' 'specialise' '[' NameTimesList? ']'
  ;

NameTimes ::= FnName Natural?;
```

```
NameTimesList ::=
  NameTimes
  | NameTimes ',' NameTimesList
  ;

CaseExpr ::=
  'case' Expr 'of' OpenBlock CaseOption+ CloseBlock;

CaseOption ::=
  Expr '=>' Expr Terminator
  ;

{- NOTE: Consider using OpenBlock CloseBlock for proofs and tactics -}
ProofExpr ::=
  'proof' OpenBlock Tactic'* CloseBlock
  ;

TacticsExpr :=
  'tactics' OpenBlock Tactic'* CloseBlock
;

Tactic' ::=
  Tactic Terminator
;


SimpleExpr ::=
  '![' Term ']'
  | '?' Name
  | % 'instance'
  | 'refl' ('{' Expr '}')?
  | ProofExpr
  | TacticsExpr
  | CaseExpr
  | FnName
  | List
  | Comprehension
  | Alt
  | Idiom
  | '(' Bracketed
  | Constant
  | Type
  | '_|_'
  | '_'
  | {- External (User-defined) Simple Expression -}
  ;

Bracketed ::=
  ')'
  | Expr ')'
  | ExprList ')'
  | Expr '**' Expr ')'
  | Operator Expr ')'
  | Expr Operator ')'
  | Name ':' Expr '**' Expr ')'
  ;


ListExpr ::=
  '[' ExprList? ']'
;

ExprList ::=
  Expr
  | Expr ',' ExprList
  ;

Alt ::= '(|' Expr_List '|)';

Expr_List ::=
  Expr'
  | Expr' ',' Expr_List
  ;

HSimpleExpr ::=
  '.' SimpleExpr
  | SimpleExpr
  ;

MatchApp ::=
  SimpleExpr '<==' FnName
  ;

UnifyLog ::=
  '%' 'unifyLog' SimpleExpr
  ;

App ::=
  'mkForeign' Arg Arg*
  | SimpleExpr Arg+
  ;

Arg ::=
  ImplicitArg
  | ConstraintArg
  | SimpleExpr
  ;
```

```
294
      ImplicitArg ::=
        '{' Name ('=' Expr)? '}'
297     ;

      ConstraintArg ::=
300     '@{' Expr '}'
        ;

303   RecordType ::=
        'record' '{' FieldTypeList '}';

306   FieldTypeList ::=
        FieldType
        | FieldType ',' FieldTypeList
309     ;

      FieldType ::=
312     FnName '=' Expr
        ;

315   TypeSig ::=
        ':' Expr
        ;
318
      TypeExpr ::= ConstraintList? Expr;

321   Lambda ::=
          '\\' TypeOptDeclList '=>' Expr
        | '\\' SimpleExprList  '=>' Expr
324     ;

      SimpleExprList ::=
327     SimpleExpr
        | SimpleExpr ',' SimpleExprList
        ;
330
      RewriteTerm ::=
        'rewrite' Expr ('==>' Expr)? 'in' Expr
333     ;

      Let ::=
336     'let' Name TypeSig'? '=' Expr 'in' Expr
      | 'let' Expr'                '=' Expr' 'in' Expr

339   TypeSig' ::=
        ':' Expr'
        ;
342
      QuoteGoal ::=
        'quoteGoal' Name 'by' Expr 'in' Expr
345     ;

      Pi ::=
348       '|'? Static? '('           TypeDeclList ')' DocComment '->' Expr
        | '|'? Static? '{'           TypeDeclList '}'            '->' Expr
        |              '{' 'auto'    TypeDeclList '}'            '->' Expr
351     |              '{' 'default' TypeDeclList '}'            '->' Expr
        |              '{' 'static'              '}' Expr'      '->' Expr
        ;
354
      ConstraintList ::=
          '(' Expr_List ')' '=>'
357     | Expr              '=>'
        ;

360   UsingDeclList ::=
        UsingDeclList'
        | NameList TypeSig
363     ;

      UsingDeclList' ::=
366     UsingDecl
        | UsingDecl ',' UsingDeclList'
        ;
369
      NameList ::=
        Name
372     | Name ',' NameList
        ;

375   UsingDecl ::=
        FnName TypeSig
        | FnName FnName+
378     ;

      TypeDeclList ::=
381       FunctionSignatureList
        | NameList TypeSig
        ;
384
      FunctionSignatureList ::=
          Name TypeSig
387     | Name TypeSig ',' FunctionSignatureList
        ;

390   TypeOptDeclList ::=
          NameOrPlaceholder TypeSig?
        | NameOrPlaceholder TypeSig? ',' TypeOptDeclList
```

```
393     ;

        NameOrPlaceHolder ::= Name | '_';
396
        Comprehension ::= '[' Expr '|' DoList ']';

399     DoList ::=
            Do
          | Do ',' DoList
402       ;

        Do' ::= Do Terminator;
405
        DoBlock ::=
          'do' OpenBlock Do'+ CloseBlock
408       ;

        Do ::=
411         'let' Name  TypeSig? '=' Expr
          | 'let' Expr'                '=' Expr
          | Name  '<-' Expr
414       | Expr' '<-' Expr
          | Expr
          ;
417
        Idiom ::= '[|' Expr '|]';

420     Constant ::=
            'Integer'
          | 'Int'
423       | 'Char'
          | 'Float'
          | 'String'
426       | 'Ptr'
          | 'Bits8'
          | 'Bits16'
429       | 'Bits32'
          | 'Bits64'
          | 'Bits8x16'
432       | 'Bits16x8'
          | 'Bits32x4'
          | 'Bits64x2'
435       | Float_t
          | Natural_t
          | String_t
438       | Char_t
          ;

441     Static ::=
          '[' static ']'
        ;
444
        Record ::=
            DocComment Accessibility? 'record' FnName TypeSig 'where' OpenBlock Constructor Terminator CloseBlock;
447
        DataI ::= 'data' | 'codata';

450     Data ::= DocComment? Accessibility? DataI FnName TypeSig ExplicitTypeDataRest?
               | DocComment? Accessibility? DataI FnName Name*   DataRest?
               ;
453
        Constructor' ::= Constructor Terminator;

456     ExplicitTypeDataRest ::= 'where' OpenBlock Constructor'* CloseBlock;

        DataRest ::= '=' SimpleConstructorList Terminator
459                | 'where'!
                   ;

462     SimpleConstructorList ::=
            SimpleConstructor
          | SimpleConstructor '|' SimpleConstructorList
465       ;

        Constructor ::= DocComment? FnName TypeSig;
468     SimpleConstructor ::= FnName SimpleExpr* DocComment?

        Overload' ::= Overload Terminator;
471
        DSL ::= 'dsl' FnName OpenBlock Overload'+ CloseBlock;

474     OverloadIdentifier ::= 'let' | Identifier;

        Overload ::= OverloadIdentifier '=' Expr;
477
        Pattern ::= Clause;

480     CAF ::= 'let' FnName '=' Expr Terminator;

        ArgExpr ::= HSimpleExpr | {- In Pattern External (User-defined) Expression -};
483
        ImplicitOrArgExpr ::= ImplicitArg | ArgExpr;

486     RHS ::= '='            Expr
              |  '?='  RHSName? Expr
              | 'impossible'
489           ;

        RHSName ::= '{' FnName '}';
```

```
492
      RHSOrWithBlock ::= RHS WhereOrTerminator
                       | 'with' SimpleExpr OpenBlock FnDecl+ CloseBlock
495                    ;

      Clause ::=                                                                  WExpr+ RHSOrWithBlock
498          |   SimpleExpr '<=='  FnName                                              RHS WhereOrTerminator
             |                     FnName ConstraintArg* ImplicitOrArgExpr*     WExpr* RHSOrWithBlock
             |   ArgExpr Operator ArgExpr                                       WExpr* RHSOrWithBlock
501          ;


504   WhereOrTerminator ::= WhereBlock  |Terminator;

      WExpr ::= '|' Expr';
507
      WhereBlock ::= 'where' OpenBlock Decl+ CloseBlock;

510   Codegen ::= 'C'
              |   'Java'
              |   'JavaScript'
513           |   'Node'
              |   'LLVM'
              |   'Bytecode'
516           ;

      Totality ::= 'partial' | 'total'
519
      StringList ::=
        String
522     | String ',' StringList
        ;

525   Directive ::= '%' Directive';

      Directive' ::= 'lib'      CodeGen String_t
528              |   'link'     CodeGen String_t
                 |   'flag'     CodeGen String_t
                 |   'include'  CodeGen String_t
531              |   'hide'     Name
                 |   'freeze'   Name
                 |   'access'   Accessibility
534              |   'default'  Totality
                 |   'logging'  Natural
                 |   'dynamic'  StringList
537              |   'language' 'TypeProviders'
                 ;

540   Provider ::= '%' 'provide' '(' FnName TypeSig ')' 'with' Expr;

      Transform ::= '%' 'transform' Expr '==>' Expr
543
      Tactic ::= 'intro' NameList?
             |   'intros'
546          |   'refine'      Name Imp+
             |   'mrefine'     Name
             |   'rewrite'     Expr
549          |   'equiv'       Expr
             |   'let'         Name ':' Expr' '=' Expr
             |   'let'         Name           '=' Expr
552          |   'focus'       Name
             |   'exact'       Expr
             |   'applyTactic' Expr
555          |   'reflect'     Expr
             |   'fill'        Expr
             |   'try'         Tactic '|' Tactic
558          |   '{' TacticSeq '}'
             |   'compute'
             |   'trivial'
561          |   'solve'
             |   'attack'
             |   'state'
564          |   'term'
             |   'undo'
             |   'qed'
567          |   'abandon'
             |   ':' 'q'
             ;
570
      Imp ::= '?' | '_';

573   TacticSeq ::=
          Tactic ';' Tactic
        | Tactic ';' TacticSeq
576     ;

      {- Open Block Close Block should match i.e. either  {}or indentation -}
579
      OpenBlock ::= '{'
                  | {- Same Indentation Level Or Greater -}
582               ;

      CloseBlock ::= '}'
585              | {- Smaller Indentation Level Than Before -}
                  ;

588   Terminator ::= ';'
                 |  {- Smaller inden than before -}
                  |  {- '|', ')' or '}' is at start of input -}
```

```
591             |  eof
                ;

594   Float_t ::= {- Float literal similar to Haskell -}
              ;

597   IName_t ::= {- Any valid identifier except keywords possibly prefixed with a namespace -}
            ;

600   Integer_t ::= {- Integer literal similar to Haskell -}
            ;

603   Char_t ::= {- Char literal similar to Haskell -}
            ;

606   Natural_t ::= {- Natural number literal i.e. [1-9]*[0-9] -}
              ;
```